<?xml version="1.0" encoding="UTF-8" ?>
 
<xsl:stylesheet version="1.0" 
        xmlns:xsl="http://www.w3.org/1999/XSL/Transform" 
        xmlns="http://www.w3.org/1999/xhtml">
    <xsl:output method="xml" indent="no"
        doctype-public="-//W3C//DTD XHTML 1.0 Strict//EN" 
        doctype-system="http://www.w3.org/TR/xhtml1/DTD/xhtml1-strict.dtd"/>

<xsl:include href="../../../templates/popup.xsl"/>

<xsl:template match="*"/>

<xsl:template match="/root">
        <xsl:apply-templates select="*"/>
</xsl:template>

<xsl:template match="report_data">
    <table>
    <xsl:apply-templates select="row"/>
    </table>
</xsl:template>

<xsl:template match="row">
    <xsl:apply-templates select="source_link"/>
    <xsl:apply-templates select="source_search"/>
    <tr>
        <td><a href="report.php?module=popularity&amp;report=hits&amp;date1={/root/date1}&amp;date2={/root/date2}&amp;by=visitor&amp;id={@VID}"><xsl:value-of select="@VID"/></a><br/><span class="small gray">(<xsl:value-of select="@VName"/>)</span></td>
        <td><a href="report.php?module=popularity&amp;report=hits&amp;date1={/root/date1}&amp;date2={/root/date2}&amp;by=session&amp;id={@SID}"><xsl:value-of select="@SID"/></a></td>
        <td><xsl:value-of select="@date"/><xsl:text> </xsl:text><xsl:value-of select="@time"/></td>
        <td style="text-align: left;">
            <a href="{@url}" target="_blank" onclick="return openPopup(this);" class="external"><xsl:value-of select="@urlHuman"/></a><br />
            <xsl:value-of select="@titleHuman"/>
        </td>
    </tr>
</xsl:template>

<xsl:template match="source_link">
    <tr>
        <td colspan="4" style="text-align: left;">↓ Переход с <a href="{@url}" target="_blank" onclick="return openPopup(this);" class="external"><xsl:value-of select="@urlHuman"/></a></td>
    </tr>
</xsl:template>

<xsl:template match="source_search">
    <tr>
        <td colspan="4" style="text-align: left;">↓ Поиск в <xsl:value-of select="@engine"/>, страница <xsl:value-of select="@page"/>, <xsl:value-of select="@textHuman"/></td>
    </tr>
</xsl:template>


</xsl:stylesheet>